eman ta zabal zazu

Universidad del País Vasco
Euskal Herriko Unibertsitatea

Ejercicios resueltos

OCW 2018: *Utilizando Mathematica como apoyo al cálculo algebraico en los grados de Ingeniería*

Tema 2. Cálculo matricial

Equipo docente del curso
Martín Yagüe, Luis
Unzueta Inchaurbe, Aitziber
Arrospide Zabala, Eneko
García Ramiro, María Begoña
Soto Merino, Juan Carlos
Alonso González, Erik

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I

EJERCICIOS DEL TEMA 2. CÁLCULO MATRICIAL

Ejercicio nº1

Enunciado

Defina y presente en forma matricial:

- a) $A = \begin{pmatrix} 1 & 0 & -2 & 1 \\ -3 & -5 & 0 & \sqrt{2} \end{pmatrix}$
- b) una matriz M_2 formada por ceros y unos obtenidos de forma aleatoria y tal que su producto con otra matriz $M_1 \in \mathbb{M}_{4\times2}(\mathbb{R})$ sea una matriz $P \in \mathbb{M}_{2\times2}(\mathbb{R})$. (Nota. Función **RandomInteger**)
- c) una matriz columna con los cuatro primeros números naturales
- d) una matriz $M \in \mathbb{M}_{3\times3}(\mathbb{R})$ cuyos elementos coinciden con el número de la fila que ocupan en la matriz
- e) una matriz diagonal que contenga en su diagonal principal los cuatro primeros números naturales
- f) una matriz diagonal que contenga en su diagonal principal los elementos de la matriz M_2 del apartado b

Resolución

La representación en forma matricial se realiza con la función **MatrixForm**

- a) $A = \begin{pmatrix} 1 & 0 & -2 & 1 \\ -3 & -5 & 0 & \sqrt{2} \end{pmatrix}$
  - Definición de la matriz

```
m1a = {{1, 0, -2, 1}, {-3, -5, 0, √2}};
```

  - En forma matricial

```
MatrixForm[m1a]
```

$$\begin{pmatrix} 1 & 0 & -2 & 1 \\ -3 & -5 & 0 & \sqrt{2} \end{pmatrix}$$

- b) una matriz M_2 formada por ceros y unos obtenidos de forma aleatoria y tal que su producto con otra matriz $M_1 \in \mathbb{M}_{4\times2}(\mathbb{R})$ sea una matriz $P \in \mathbb{M}_{2\times2}(\mathbb{R})$. (Nota. Consúltese la ayuda de la función **RandomInteger**)
  - Definición de la matriz: $M_2 \in \mathbb{M}_{2\times4}(\mathbb{R})$ de forma que $M_2 \cdot M_1 = P \in \mathbb{M}_{2\times2}(\mathbb{R})$

```
m21b = RandomInteger[1, {2, 4}]
```

```
{{1, 1, 0, 0}, {0, 0, 1, 1}}
```

  - En forma matricial

```
MatrixForm[m21b]
```

$$\begin{pmatrix} 1 & 1 & 0 & 0 \\ 0 & 0 & 1 & 1 \end{pmatrix}$$

- c) una matriz columna con los cuatro primeros números naturales
  - Definición de la matriz

```
m1c = {1, 2, 3, 4};
```

  - En forma matricial

```
MatrixForm[m1c]
```

$$\begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}$$

- d) una matriz $M \in \mathbb{M}_{3\times 3}(\mathbb{R})$ cuyos elementos coinciden con el número de la fila que ocupan en la matriz
  - Definición de la matriz

```
m1d = Table[i, {i, 3}, {j, 3}]
```

```
{{1, 1, 1}, {2, 2, 2}, {3, 3, 3}}
```

  - En forma matricial

```
MatrixForm[m1d]
```

$$\begin{pmatrix} 1 & 1 & 1 \\ 2 & 2 & 2 \\ 3 & 3 & 3 \end{pmatrix}$$

- e) una matriz diagonal que contenga en su diagonal principal los cuatro primeros números naturales
  - Definición de la matriz

```
m1e = DiagonalMatrix[m1c];
```

  - En forma matricial

```
MatrixForm[m1e]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 \\ 0 & 0 & 3 & 0 \\ 0 & 0 & 0 & 4 \end{pmatrix}$$

- f) una matriz diagonal que contenga en su diagonal principal los elementos de la matriz M_2 del apartado b
  - Definición de la matriz

```
m1f = DiagonalMatrix[Flatten[m21b]];
```

  - En forma matricial

```
MatrixForm[m1f]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Ejercicio nº2

Enunciado

Defina, cada una de forma diferente, tres matrices tales que $A, B \in \mathbb{M}_{3\times 3}(\mathbb{R})$ y $C \in \mathbb{M}_{2\times 3}(\mathbb{R})$. Determine:

- a) $2A - 3B$
- b) $C(2A - 3B)$
- c) si las matrices A y B conmutan
- d) el producto de las matrices A y C
- e) C^T
- f) que se verifica $(A+B)^T = B^T \cdot A^T$
- g) $det(A)$ y $det(B)$
- h) $A^n \quad \forall n \in \mathbb{N}$
- i) si las matrices A y B son regulares y, en caso afirmativo, calcular A^{-1} y B^{-1}

Resolución

Definición de las tres matrices:

```
a = RandomInteger[1, {3, 3}]
```

```
{{0, 1, 0}, {0, 0, 1}, {0, 0, 1}}
```

```
b = {{1, 0, 1}, {0, 1, 0}, {0, 0, 1}};
```

```
c = Table[i * j, {i, 1, 2}, {j, 1, 3}]
```

```
{{1, 2, 3}, {2, 4, 6}}
```

- a) $2A - 3B$

```
m2a = 2 * a - 3 * b
```

```
{{-3, 2, -3}, {0, -3, 2}, {0, 0, -1}}
```

- b) $C(2A - 3B)$

```
m2b = c.m2a
```

```
{{-3, -4, -2}, {-6, -8, -4}}
```

- c) si las matrices A y B conmutan

```
a.b == b.a
```

```
False
```

- d) el producto de las matrices A y C

```
m2d = c.a (* única forma compatible con la definción del producto de matrices*)
```

```
{{0, 1, 5}, {0, 2, 10}}
```

```
a.c (* las matrices no tienen órdenes compatibles con el producto planteado: mensaje de error *)
```

```
... Dot: Tensors {{0, 1, 0}, {0, 0, 1}, {0, 0, 1}} and {{1, 2, 3}, {2, 4, 6}} have incompatible shapes.
```

```
{{0, 1, 0}, {0, 0, 1}, {0, 0, 1}}.{{1, 2, 3}, {2, 4, 6}}
```

- e) C^{T}

```
m2e = Transpose[c] // MatrixForm
```

$$\begin{pmatrix} 1 & 2 \\ 2 & 4 \\ 3 & 6 \end{pmatrix}$$

- f) que se verifica $(A\cdot B)^{\mathrm{T}} = B^{\mathrm{T}}\cdot A^{\mathrm{T}}$

```
Transpose[a.b] == Transpose[b].Transpose[a]
```

```
True
```

- g) $det(A)$ y $det(B)$

```
Det[a] (* la matriz a no es regular *)
```

```
0
```

```
Det[b] (* la matriz b sí es regular *)
```

```
1
```

- h) $A^n \quad \forall n \in \mathbb{N}$

```
MatrixPower[b, n] // Simplify // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & n \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

- i) si las matrices A y B son regulares y, en caso afirmativo, calcular A^{-1} y B^{-1}
  - sólo es regular la matriz b

```
Inverse[b] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & -1 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Ejercicio nº3

Enunciado

Defina una matriz $B = \left(b_{\mathrm{ij}}\right) \in \mathbb{M}_{3\times 6}(\mathbb{R})$.

Extraiga, de dos formas diferentes:

- a) la fila tres
- b) el elemento b_{25}
- c) la columna cuatro
- d) la submatriz $B_1 = \begin{pmatrix} b_{12} & b_{13} & b_{15} \\ b_{32} & b_{33} & b_{35} \end{pmatrix} \in \mathbb{M}_{2\times 3}(\mathbb{R})$

Determine:

- e) el rango de B, $r(B)$
- f) una matriz equivalente por filas a B

Resolución

Definición de la matriz:

```
b3 = Table[i*j, {i, 1, 3}, {j, 1, 6}]
```

```
{{1, 2, 3, 4, 5, 6}, {2, 4, 6, 8, 10, 12}, {3, 6, 9, 12, 15, 18}}
```

```
MatrixForm[b3]
```

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 & 6 \\ 2 & 4 & 6 & 8 & 10 & 12 \\ 3 & 6 & 9 & 12 & 15 & 18 \end{pmatrix}$$

- a) la fila tres

```
b3[[3]]
```

```
{3, 6, 9, 12, 15, 18}
```

```
Part[b3, 3]
```

```
{3, 6, 9, 12, 15, 18}
```

- b) el elemento b_{25}

```
b3[[2, 5]]
```

```
10
```

```
Part[b3, 2, 5]
```

```
10
```

- c) la columna cuatro

```
b3t = Transpose[b3]
```

```
{{1, 2, 3}, {2, 4, 6}, {3, 6, 9}, {4, 8, 12}, {5, 10, 15}, {6, 12, 18}}
```

```
MatrixForm[b3t]
```

$$\begin{pmatrix} 1 & 2 & 3 \\ 2 & 4 & 6 \\ 3 & 6 & 9 \\ 4 & 8 & 12 \\ 5 & 10 & 15 \\ 6 & 12 & 18 \end{pmatrix}$$

```
b3t[[4]]
```

```
{4, 8, 12}
```

```
Part[b3t, 4]
```

```
{4, 8, 12}
```

- d) la submatriz $B_1 = \begin{pmatrix} b_{12} & b_{13} & b_{15} \\ b_{32} & b_{33} & b_{35} \end{pmatrix} \in \mathbb{M}_{2\times 3}(\mathbb{R})$

```
b3[[{1, 3}, {2, 3, 5}]]
```

```
{{2, 3, 5}, {6, 9, 15}}
```

```
Part[b3, {1, 3}, {2, 3, 5}]
```

```
{{2, 3, 5}, {6, 9, 15}}
```

- e) el rango de B, $r(B)$

```
MatrixRank[b3]
```

```
1
```

- f) una matriz equivalente por filas a B

```
RowReduce[b3] // MatrixForm
```

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 & 6 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

Ejercicio nº4

Enunciado

Hallar, en función de los valores del parámetro $m \in \mathbb{R}$, el rango de la matriz:

$$A = \begin{pmatrix} 1 & m & 1 & -1 \\ 0 & 1 & m-1 & 0 \\ 1 & 1 & m & -1 \end{pmatrix}$$

Resolución

- Introducción de la matriz 3×4

```
a4 = {{1, m, 1, -1}, {0, 1, m - 1, 0}, {1, 1, m, -1}};
```

- Cálculo de los cuatro menores de orden tres: $r(A)\leq3$

```
men = Minors[a4, 3]
```

$\{\{-m + m^2, 0, 0, m - m^2\}\}$

- Conversión de la salida anterior en una lista cuyos elementos son los menores calculados

```
l = Flatten[men]
```

$\{-m + m^2, 0, 0, m - m^2\}$

- Cálculo de los valores del parámetro m que anulan todos los menores

```
sol = Solve[Table[l[[i]] == 0, {i, Length[l]}], m]
```

$\{\{m \to 0\}, \{m \to 1\}\}$

- Sustitución de los valores del parámetro m (dos en este caso) en la matriz original y cálculo de su rango

```
MatrixRank[a4 /. sol[[1]]]
```

```
2
```

```
MatrixRank[a4 /. sol[[2]]]
```

```
2
```

- Solución: $\boxed{\begin{array}{l} r(A) = 2 \quad \text{si } m = 0 \vee m = 1 \\ r(A) = 3 \quad \text{si } m \neq 0 \wedge m \neq 1 \end{array}}$

Ejercicio nº5

Enunciado

Sea la matriz $P = \begin{pmatrix} 1 & 1 \\ 0 & 1 \end{pmatrix} \in \mathbb{M}_{2\times 2}(\mathbb{R})$.

Calcular : $\sum_{k=1}^{n} P^k \quad \forall\, n \in \mathbb{N}$

Resolución

- Introducción de la matriz 2×2

```
p5 = {{1, 1}, {0, 1}}
```

```
{{1, 1}, {0, 1}}
```

```
MatrixForm[p5]
```

$$\begin{pmatrix} 1 & 1 \\ 0 & 1 \end{pmatrix}$$

- Potencia k-ésima

```
p5n = MatrixPower[p5, k] // Simplify
```

```
{{1, k}, {0, 1}}
```

```
MatrixForm[p5n]
```

$$\begin{pmatrix} 1 & k \\ 0 & 1 \end{pmatrix}$$

- Cálculo del sumatorio

```
s[n_] = Sum[p5n, {k, 1, n}] // Simplify; MatrixForm[s[n]]
```

$$\begin{pmatrix} n & \frac{1}{2} n (1+n) \\ 0 & n \end{pmatrix}$$

- Comprobación con el método de inducción:
  - cuando $n = 1$:

```
s[1] == p5
```

```
True
```

  - se supone que el sumatorio se verifica cuando $n = k$:

```
s[k] // MatrixForm
```

$$\begin{pmatrix} k & \frac{1}{2} k (1+k) \\ 0 & k \end{pmatrix}$$

  - se compueba que el sumatorio se verifica cuando $n = k+1$:

```
s[k] + MatrixPower[p5, k + 1] == s[k + 1] // Simplify
```

```
True
```